

Teil 3 Berechnung von Feldern

1 Stationäres elektrisches Strömungsfeld

1.1 Elektrischer Widerstand eines Quaders

Der elektrische Widerstand eines Quaders mit den Abmessungen Länge $l = 100$ mm, Breite $b = 20$ mm und Höhe $h = 10$ mm in Längsrichtung kann durch die Bemessungsgleichung

$$R = \frac{l}{\gamma bh}$$

ermittelt werden. Dabei ist γ die spezifische elektrische Leitfähigkeit des verwendeten Werkstoffs. Für Kupfer beträgt sie $\gamma_{\text{Cu}} = 56 \text{ m}/(\Omega \text{ mm}^2)$. Damit ergibt sich für die angenommenen Werte ein Widerstand von $R = 8,9285 \mu\Omega$.

Dieser Wert soll mit ANSYS nachgerechnet werden. Dazu wird ein Quader modelliert und an einer Seite das Potenzial 0 erzwungen und an der anderen Seite das Potenzial gekoppelt und ein Strom eingespeist. Der Widerstand ergibt sich dann aus dem Quotienten der entstehenden Spannung und dem eingespeisten Strom.

```
!Ermittlung der el. Widerstandes eines Quaders
!-----Parameter-----
lang=100e-3      !Länge des Quaders
hoch=10e-3       !Höhe des Quaders
breit=20e-3      !Breite des Quaders
strom=1000       !eingepägter Strom
!-----
/prep7
```

Hier wird ein räumliches Modell der gesamten Anordnung (ohne Ausnutzung von Symmetrien) erstellt. Als Elementtyp eignet sich `solid69`.

```
et,1,solid69
mp,rsvx,1,1/57e6      !Kupfer
!----Geometrie-----
block,0,breit,0,hoch,0,lang
!----Vernetzung-----
esize,,1            !nur ein Element
vmesh,1
```

Abb. 1.1-1 zeigt das vernetzte Volumen.

Alle Knoten bei $z = \text{lang}$ werden bezüglich des `volt`-Freiheitsgrades gekoppelt und in einen dieser Knoten der Strom eingespeist.




Abb. 1.1-1 Generiertes Volumen

```
!-----Lasten-----
nselect,s,loc,z,lang
cpl,1,volt,all
*get,node1,node,,num,min
f,node1,amps,strom
```

Auf der anderen Seite werden alle Knoten auf das Potenzial 0 gelegt.

```
nselect,s,loc,z,0
d,all,volt,0
allsel
```

Damit kann die Lösung angefordert werden.

```
!-----Lösung-----
/solu
solve
!-----Auswertung-----
/post1
```

Zur Kontrolle wird die Potenzialverteilung betrachtet. (Dazu muss die Geometrie etwas gedreht werden.)

```
plns,volt
```

Die Potenzialverteilung in Abb. 1.1-2 entspricht den Erwartungen.

An dem vorher ermittelten Knoten (an dem der Strom eingespeist wurde), wird das Potenzial abgefragt und für folgende Berechnung verwendet.




Abb. 1.1-2 Potenzialverteilung

```
*get,spannung,node,node1,volt
widerst=spannung/strom
```

Im ANSYS-Output lässt sich der Wert

```
PARAMETER WIDERST .89285E-5
```

ablesen.

1.2 Erwärmung eines Leiters bei Stromfluss

Ein sehr langer rechteckiger Leiter aus Stahl mit den Abmessungen Breite $b = 20$ mm und Höhe $h = 10$ mm wird in Längsrichtung von einem Strom von 5000 A durchflossen. Von diesem Leiter wird ein Stück der Länge $l = 1$ mm betrachtet. Durch die Enden wird keine Wärmeleitung angenommen.

Der Leiter erwärmt sich durch den Stromfluss. Der Erwärmungsvorgang soll als Funktion des Ortes und der Zeit untersucht werden. Dabei ist die Wärmeabgabe an der Oberfläche durch Konvektion zu berücksichtigen.

Wegen der Symmetrie braucht nur ein Viertel untersucht zu werden.

Programm

```
!Elektrisches Strömungsfeld
!transient thermisch

!-----Parameter-----
lang=1e-3           !Länge des Quaders
hoch=10e-3          !Höhe des Quaders
breit=20e-3         !Breite des Quaders
strom=2000          !eingepägter Strom
zeitende=60         !1 Minute
```

Der Elementtyp, der für diese Aufgabe geeignet ist, muss dreidimensionale Elemente und die Freiheitsgrade `volt` und `temp` besitzen. Dafür wären die Typen `solid5`, `solid69` und `solid226,110` geeignet. Hier wird `solid226,10` benutzt.

```

/prep7
et,1,solid226,110
mp,rsvx,1,1/10e6      !el. Leitwert
mp,kxx,1,52          !Wärmeleitwert
mp,c,1,420           !spez. Wärmekapazität
mp,dens,1,7860       !Dichte

```

Zur Generierung der Geometrie wird ein Quader vereinbart. Die Breite des Leiters liegt in X-Richtung, die Höhe in Y-Richtung und die Länge in Z-Richtung.

```

!-----Geometrie----
block,0,breit/2,0,hoch/2,0,lang

```

Als Maschenweite für die Vernetzung wird ein Achtel der Höhe vereinbart.

```

!-----Vernetzung---
esize,hoch/8
vmesh,all

```

Abb. 1.2-1 zeigt das vernetzte Volumen.




Abb. 1.2-1 Vernetztes Volumen

Zum Aufbringen der elektrischen Randbedingungen werden alle Knoten bei $Z = 0$ selektiert und hinsichtlich des Freiheitsgrades Potentials gekoppelt. In einem dieser Knoten wird der Strom (ein Viertel des Gesamtstroms, da nur ein Viertel des Leiters modelliert wurde) eingespeist. Die gegenüberliegenden Knoten (bei $Z = \text{lang}$) werden auf Bezugspotenzial gelegt.

```

!---Randbedingungen--elektrisch-----
nselect,s,loc,z,0
cp,1,volt,all          !Kopplung des Potentials
*get,nodel,node,,num,min !
f,nodel,amps,Strom/4
nselect,s,loc,z,lang
d,all,volt,0

```

Zum Aufbringen der thermischen Randbedingungen werden alle Knoten an den Außenflächen des Leiters selektiert und dort der Wärmeübergangskoeffizient (hier sehr hoch angenommen) und die Temperatur des umgebenden Mediums als Flächenlast aufgebracht.

```

!-----Randbedingungen---thermisch-----
nselect,s,loc,x,breit/2
nselect,a,loc,y,hoch/2
sf,all,conv,1000,20
tunif,20                !Umgebungstemperatur

```

Zu Beginn der Lösung werden alle Selektierungen rückgängig gemacht und die transiente Lösung vereinbart. Das vorgegebene Zeit-Ende soll in 20 Rechenschritten erreicht werden.

```

allselect
!---Lösung-----
/solu
antype,trans
kbc,1                !keine Rampe
time,zeitende
schritte=20
deltim,zeitende/schritte
outres,all,all      !alle Lösungsschritte
solve

```

Wenn neben dem Aufheizvorgang auch der Abkühlvorgang berechnet werden soll, kann einfach nach Aufheben aller Lasten, die mit dem `f`-Kommando aufgebracht waren (in diesem Beispiel war das der Strom) weitergerechnet werden.

```

!----Abkühlung----
fdelete,all,all
time,zeitende*2
solve

```

In der Auswertung kann der Verlauf der Temperatur über der Zeit in der Leitermitte und an einer Außenkante mit folgenden Anweisungen dargestellt werden:

```

!---Auswertung-----
/post26          !Zeitprozessor

```

```

nsol,2,node(0,0,lang/2),temp
nsol,3,node(breit/2,hoch/2,lang/2),temp
plvar,2,3

```

Abb. 1.2-2 zeigt das Ergebnis.




Abb. 1.2-2 Temperaturen über der Zeit

Mit Hilfe des anderen Postprozessors kann man sich die Temperaturen in der Schnittfläche ansehen. Dazu muss ein bestimmter Zeitpunkt mit dem `set`-Befehl (z.B. 60 s) vereinbart werden.

```

/post1
set,,,,,60 !60 Sekunden
plnsol,temp

```

Abb. 1.2-3 zeigt das Ergebnis.




Abb. 1.2-3 Temperaturverteilung nach 60 s

1.3 Punktschweißlinse

Ausbildung der Schweißlinse bei der Punktschweißung

Aufgabenstellung

Zwei Elektroden aus Kupfer pressen zwei zu verschweißende Bleche zusammen. Durch die Anordnung wird kurzzeitig ein hoher Strom geschickt, der die Bleche an der Berührungsstelle zum Schmelzen bringt. Nach Abschalten des Stromes bleiben die Bleche noch eine zeitlang unter dem Druck der Elektroden, bis der aufgeschmolzene Bereich wieder erstarrt ist.

Es soll hier das Prinzip der Aufheizung und der Ausbildung der Schweißlinse untersucht werden. Dabei soll (neben vielen anderen Vereinfachungen) der an sich wichtige elektrische Übergangswiderstand zwischen den Elektroden und den zu verschweißenden Teilen und zwischen den zu verschweißenden Teilen unberücksichtigt bleiben.

Aufbereitung

Wie aus dem Bild zu erkennen ist, ist der innere, hier interessierende Teil der Anordnung rotationssymmetrisch, so dass zweidimensional gerechnet werden kann. Weiterhin besteht in Längsrichtung eine Symmetrie. Der für die Berechnung notwendige Teil ist also ein Viertel der Schnittfläche.




Abb. 1.3-1 Geometrie einer Punktschweißanordnung

Programm

Das dazu erforderliche Programm beginnt mit Kopf und Variablenvereinbarung:

```

!Punktschweißlinse
!rotationssymmetrisch
!Erwärmung
radelu=3e-3           !Radius Elektrode unten
radelo=radelu*2.5    !Radius Elektrode oben
hoel=4e-3            !Höhe Elektrode
radble=20e-3         !Radius Blech
hoble=1e-3           !Höhe Blech
rocu=1/56e6          !spez. Widerstand der Kupferelektroden
lambdafe=84          !Wärmeleitfähigkeit der Bleche
lambdacu=393         !Wärmeleitfähigkeit der Elektroden
alfa=500             !Wärmeübergangszahl
tung=20              !Umgebungstemperatur
dichtefe=7860        !Dichte der Bleche
dichtecu=8930        !Dichte der Elektroden
cfe=460              !Wärmekapazität der Bleche
ccu=383              !Wärmekapazität der Elektroden
strom=13500          !Schweißstrom

```

```
zeitende=0.300    !Schweißzeit
```

Nach Aufruf des Preprozessors kann die Elementtyp- und Materialvereinbarung erfolgen.

```
/prep7
et,1,plane67,,,1
mp,kxx,1,lambdafe !Blech
mp,c,1,cfe
mp,dens,1,dichtefe
mp,kxx,2,lambdacu !Elektrode
mp,rsvx,2,rocu
mp,c,2,ccu
mp,dens,2,dichtecu
```

Beispielhaft soll hier der spezifische elektrische Widerstand der Blechmaterials temperaturabhängig angenommen werden. (Natürlich sind eigentlich alle Materialeigenschaften temperaturabhängig. Auch die Schmelzwärme des Materials soll hier unberücksichtigt bleiben.) Zur Kontrolle wird die Abhängigkeit grafisch in Abb. 1.3-2 dargestellt.

```
mptemp,1,0,200,600,1000,1200,1600
mp,rsvx,1,.21e-6,-2.36e-10,4.13e-12,-4.09e-15,1.15e-18
/yrange,0,1.2e-6
/axlab,y,spez.Widerstand
/axlab,x,Temperatur
/grid,1
/gthk,curve,2
mpplot,rsvx,1
```




Abb. 1.3-2 Temperaturabhängigkeit der spez. el. Widerstands der Bleche

Zur Geometriegenerierung der Elektrode wird die Fläche hier über Keypoints definiert. Das Blech wird aus Vernetzungsgründen durch zwei Rechtecke gebildet. Alle Teilflächen werden zusammengeklebt.

```

!----Geometrie----
k,1,0,hoble
k,2,radelu,hoble
k,3,radelo,hoble+hoel
k,4,0,hoble+hoel
a,1,2,3,4                !Elektrode
rectng,0,radelu,0,hoble  !Blech innen
rectng,radelu,radble,0,hoble !Blech außen
aglu,all
!A1:Elektrode A4:Blech innen A5:Blech außen

```



Abb. 1.3-3 Geometrie nach Vernetzung

Es folgen die für die Vernetzung erforderlichen Anweisungen

```

!---Vernetzung---
eshape,2
esize,hoble/4
mat,1
amesh,4                !Blech innen
lsel,s,loc,x,radelu+(radble-radelu)/2
lesize,all,,,8,8
amesh,5                !Blech außen
!----Elektrode----
lsel,s,loc,y,hoble+hoel/2
lsel,r,loc,x,radelu,radelo
lesize,all,,,6,6
lsel,s,loc,y,hoble+hoel/2
lsel,r,loc,x,0
lesize,all,,,6,1/6
mat,2
amesh,1

```

Für die Stromeinprägung werden die Knoten am oberen Elektrodenrand gekoppelt und in einen von ihnen der Strom eingepreßt. Den Knoten zwischen den Blechen (Symmetrieebene) wird das Potenzial 0 zugewiesen.

```

!----Randvorgaben-----
nselect,s,loc,y,hoble+hoel
cp,1,volt,all
*get,n1,node,,num,min
f,n1,amps,strom
nselect,s,loc,y,0
d,all,volt,0
!-----

```

Zur Berücksichtigung der inneren Wasserkühlung der Elektroden wird ein Knoten auf Umgebungstemperatur gehalten.

```

nselect,s,loc,x,0
nselect,r,loc,y,hoble+hoel
d,all,temp,tung
!-----

```

Zur Berücksichtigung der Wärmeabgabe an die Umgebung durch Konvektion werden die infrage kommenden Knoten selektiert und ihnen eine Wärmeübergangszahl und die Temperatur des umgebenden Mediums zugewiesen. Alle Knoten werden auf Umgebungstemperatur gesetzt.

```

nselect,s,ext
nselect,u,loc,y,0
nselect,u,loc,y,hoble+hoel
nselect,u,loc,x,0
nselect,u,loc,x,radble
sf,all,conv,alfa,tung
nselect,all
tunif,tung
!-----

```

Nach Aufhebung aller Selektionen kann die Lösung angefordert werden. Um die zeitlichen Verläufe darstellen zu können, muss transient gerechnet werden. Wegen der Nichtlinearität der Materialeigenschaft empfiehlt sich die Verwendung der automatische Schrittweitensteuerung.

```

allsel
!-----
/solu
antype,trans
kbc,1           !keine Rampe
time,zeitende
nsubst,50,200,20 !Vorgabewerte
outres,,all
autots,on       !automatische Schrittweitensteuerung
solve

```

Für eine erste Auswertung soll der Temperaturverlauf in Schweißlinsenmitte und am unteren Elektrodenrand dargestellt werden.

```
!-----
/post26
nsol,2,node(0,0,0),temp,,Mitte !Linsenmitte
nsol,3,node(radelu,hoble,0),temp,,Rand !Elektrodenrand
/yrange,0,2000
/axlab,y,Temperatur [C]
/axlab,x,Zeit
plvar,2,3
/eof !Programmende bei automatischem Durchlauf
!-----
```




Abb. 1.3-4 Temperaturverlauf an zwei ausgewählten Stellen

Um den Verlauf der Temperaturentwicklung (in 10 Zeitschritten) zu betrachten, dient folgende Sequenz. Durch die Festlegung mit dem `/contour`-Befehl wird erreicht, dass der aufgeschmolzene Bereich (über 1500 °C) hell dargestellt wird.

```
/post1
/cont,,9,20,,1500 !Temperatur zwischen 20 und 1500 °C
*do,i,1,10
set,,,,,zeitende/10*i
plnsol,temp
*enddo
!-----
```

Bisher wurde nur das berechnete Viertel der Schnittfläche dargestellt. Möchte man die gesamte Schnittfläche zeigen, muss das Ergebnis mehrfach gespiegelt werden.

Für den Endzustand machen das die folgenden Anweisungen. Sie lassen sich natürlich auch in die obige Sequenz einbauen.

```

set,last
/view,,0,0,1
/focus,,0,0,0      !Darstellungsmittelpunkt
/dist,,radble*0.5  !Betrachtungsentfernung
plnsol,temp
/noerase           !Fensterlöschung aus
/view,,0,0,-1
plnsol,temp
/angle,,180       !Drehung der Darstellung
plnsol,temp
/view,,0,0,1
plnsol,temp
/erase            !Fensterlöschung ein
/angle,,0         !Rücksetzen

```

Eleganter für eine 3D-Teilschnitt-Darstellung geht dies auch mit:

```
/expand,27,axis,full,,10,2,polar,half,180
```




Abb. 1.3-5 Temperaturverteilung nach 0,3 s

Der halbe (wegen der Symmetrie) Spannungsbedarf der Anordnung wird durch folgende Anweisungen in Abb. 1.3-6 wiedergegeben.

```

/post26
nsol,2,n1,volt,
/yrange,0,0.5
/axlab,y,Spannung [V]
/axlab,x,Zeit
plvar,2

```




Abb. 1.3-6 Spannungsbedarf der halben Anordnung

Die vorstehende Betrachtung kann natürlich nur das Prinzip der Modellierung aufzeigen. Für eine Anwendung zur Untersuchung des realen Vorgangs müssen wesentlich mehr Abhängigkeiten (Nichtlinearitäten der Materialeigenschaften, Schmelzwärme des Materials, Wärmeabgabe durch Strahlung, Kontaktwiderstände) berücksichtigt werden. Hat man dann aber ein die Realität hinreichend widerspiegelndes Modell, kann eine Verfahrensdatierung und eine Fehlerursachenermittlung am Rechner oft besser als im Experiment erfolgen.

1.4 Elektroerosion

Fußpunktverdampfung

Bei der Elektroerosion bildet sich eine elektrische Entladung zwischen zwei Elektroden aus. An der Werkstückelektrode kommt es im Bereich des Fußpunktes dieser Entladung zu einer gewünschten Erwärmung bis zur Aufschmelzung und Verdampfung des Materials. Damit wird ein lokaler Abtrag bewirkt. An der Werkzeugelektrode ist dieser Vorgang unerwünscht. Er führt zum Verschleiß der Elektrode.

Die Auswirkungen des Energieeintrags an der Werkstückoberfläche sollen betrachtet werden. Der Kanal der sich ausbildenden elektrischen Entladung kann zylinderförmig angenommen werden. In dessen Fußpunkt wird ein Teil der Entladungsenergie in Wärme umgesetzt. Diese ergibt sich aus dem Katodenfall \cdot Strom. Zielstellung der Betrachtung ist die Form des sich ausbildenden Kraters.

Im ersten Schritt soll eine grobe Orientierungsrechnung durchgeführt werden.

Am Programmanfang werden die Parameter vereinbart. Die Säule der Entladung beginnt mit einem kleinen Radius (`radanf`). Danach breitet sie sich mit der Radiuswachstumsgeschwindigkeit (`radwa`) für die Dauer des Stromwachstums aus. Der Strom beginnt bei Null und steigt linear mit dem Stromwachstum (`strwa`) bis zu seinem Endwert (`strend`). Für die Geometriegenerierung wird ein mittlerer Radius (`radmi`) vereinbart. Bis zu ihm soll relativ eng vernetzt werden. Der gesamte zu untersuchende Bereich soll bis zu maximalen Radius (`radma`) gehen.

```

!Funken auf eine Platte
!rotationssymmetrisch im Schnitt
!einfache Geometrie
!mit Enthalpie
radanf=1e-6    !Radius am Anfang
strend=40     !Strom-Ende
pulsd=2e-6    !Puls-Dauer
radwa=80      !Radius-Wachstum [um/us]
strwa=200e6   !Strom-Wachstum [A/s]
radmi=25e-6   !Radius-mittlerer
radma=50e-6   !Radius-Maximum
fupusp=14     !Fußpunktspannungsabfall

/prep7
et,1,plane55,,1
mp,kxx,1,84
mp,dens,1,7860
mp,temp,1,300,1800,1810,2500,2800,3000
mpdata,enth,1,1,9.8e8,6.5e9,8.7e9,1.2e10,6.2e10,6.3e10
mplot,enth,1

```




Abb. 1.4-1 Enthalpie (Steilheit verringert)

Diese Nichtlinearität kann zu Konvergenzproblemen der auszuführenden Berechnungen führen. Um eine orientierende Berechnung ausführen zu können, kann die folgende Vereinfachung des Enthalpieverlaufes vorgenommen werden.

```
mptemp,1,300,2800,5000
mpdata,enth,1,1,9.8e8,6.2e10,6.21e10
```

Die Geometrie des zu untersuchenden Bereichs wird durch zwei Viertel-Kreisflächen gebildet.

```
!-----Geometrie-----
pcirc,0,radmi,0,-90
pcirc,0,radma,0,-90
aovlap,all
!A1:innen A3:außen
```

Der innere Teil wird relativ eng ($1\mu\text{m}$) mit gleichmäßiger Maschenweite vernetzt.

```
esize,1e-6
mshape,0,2d
mshkey,1
amesh,1
```

Der äußere Teil wird grob vernetzt.

```
csys,1
lsel,s,loc,x,radmi,radma
lsel,u,loc,x,radmi
lsel,u,loc,x,radma
lesize,all,,,6,6
esize
amesh,3
```

Die Berechnung erfolgt transient. Die Temperatur wird als absolute aufgefasst. Deshalb wird zu Beginn alles auf 300 K gesetzt.

```
tunif,300
/solu
antype,trans
kbc,1
time,1e-12
nsubst,1
solve      !Startwert
```

Die Berechnung des Endradius des Entladungskanals erfolgt aus den Vorgaben. Die Zeitschrittweite ergibt sich aus den Wachstumsgeschwindigkeiten. In den Anstieg müssen hinreichend viele Punkte gelegt werden. Damit ergibt sich die Anzahl der Berechnungsschritte.




Abb. 1.4-2 Vernetzte Fläche

```
radend=strend/strwa*radwa
sw=50e-9      !Schrittweite
schritte=pulsd/sw
```

Für jeden Berechnungsschritt wird der Strom und der aktuelle Fußpunktradius ermittelt.

```
*do,i,1,schritte
zeit=sw*i
!-----Strom-----
strom=strwa*zeit
*if,strom,gt,strend,then
strom=strend
*endif
!-----Radius-----
rad=radanf+radwa*zeit
*if,rad,gt,radend,then
rad=radend
*endif
```

Die Wärmestromdichte, die mit dem `sf`-Befehl übergeben wird, kann aus Strom, Fußpunktspannungsabfall und Fläche ermittelt werden.

```
!----Fläche-----
fla=3.1415*rad*rad
```

```

nset,s,loc,y,0
nset,r,loc,x,0,rad
sf,all,hflux,strom*fupusp/fla*1
nset,all

```

Für jeden der Schritte werden Bedingungen vorgegeben.

```

time,zeit
autots,on      !automatische Schrittweitensteuerung
neqit,50
nsubst,10,200,2
solve
*enddo

```

Die Auswertung als Animation erfolgt mit folgenden Anweisungen. Mit dem `/contour`-Befehl wird der Temperaturbereich zwischen 300 und 2900 K durch entsprechende Farben dargestellt. Bereiche mit einer Temperatur über 2900 K, die verdampftes Material enthalten, werden grau ausgewiesen.

```

/post1
/plopts,minm,0
/cont,,300,200,2900
!/expand,27,axis,full,,10,1,polar,,180
!-----
*do,i,1,schritte
set,i,last
plns,temp
/wait,0.2
*enddo

```



Abb. 1.4-3 Temperaturverteilung

Es entsteht ein Krater von ca. 30 μm Durchmesser und ca. 3 μm Tiefe.

1.5 PTC-Keramik

Gesinterte Materialien aus Bariumkarbonat und Titanoxid haben einen positiven Temperaturkoeffizienten (Kaltleiter), d.h. ihr spezifischer elektrischer Widerstand steigt mit zunehmender Temperatur. Damit lassen sich günstig Heizer ohne zusätzliche Mess- und Stellglieder aufbauen. Anwendung finden diese als Heizelement z.B. in Haartrocknern, in Spiegelheizungen im PKW, zur Temperaturmessung und als Überlastschutz.

Eine der üblichen Bauformen ist eine Scheibe (umgangssprachlich auch Pille genannt). Die gegenüberliegenden Flächen werden elektrisch leitfähig gemacht und kontaktiert. Beim Anlegen einer Spannung an die Kontakte fließt ein Strom, der das Material erwärmt. Infolge der Temperaturerhöhung steigt der Widerstand (nichtlinear) an und begrenzt den Strom. Damit stellt sich (materialabhängig) eine bestimmte Temperatur und (bauformabhängig) ein bestimmter Strom ein. Wird Wärme abgeführt, sinkt die Temperatur und damit sinkt der Widerstand und damit steigt der Strom und erhöht die zugeführte elektrische Energie, bis sich wieder das Gleichgewicht einstellt.




Abb. 1.5-1 Geometrie (Prinzip)

Für die Bildung eines Modells ist die Kenntnis der Abhängigkeit des spezifischen elektrischen Leitwerts von der Temperatur und der Werte für Wärmeleitfähigkeit und Wärmekapazität notwendig.

Hier soll eine solche PTC-Pille in ruhender Luft betrachtet werden. Da die Anordnung rotationssymmetrisch ist, kann zweidimensional axialsymmetrisch gerechnet werden.

Programm

Das dazu erforderliche Programm beginnt mit Kopf und Variablenvereinbarung

```
!PTC-Element
!Pille in ruhender Luft
radpil=2.5e-3      !Radius Pille
dipil=2e-3        !Dicke Pille
!-----
/prep7
```

```
et,1,plane67,,,1 !rotationssymmetrisch
```

```
!-----PTC-Material-----
mp,kxx,1,2.2 !Wärmeleitfähigkeit
mp,dens,1,2400 !Dichte
mp,c,1,550 !spez. Wärmekapazität
```

Der nichtlineare Zusammenhang zwischen spezifischer elektrischer Leitfähigkeit und Temperatur kann (insbesondere wenn er aus einer Messung stammt) über eine Tabelle vereinbart werden. Da ANSYS die Eingabe über den spezifischen Widerstand erwartet, kann dieser direkt eingegeben werden. Zunächst werden die Tabellenwerte für die Temperatur definiert. Im Beispiel sind es 29. Mit jeder Programmzeile können 6 Werte festgelegt werden. Der jeweils erste Parameter ist die Startnummer. (Für äquidistante Werte wäre das einfacher mittels `mtpgen` möglich.)

```
!---spez. Widerstand über Tabelle-----
mptemp,1,-50,30,39,50,60,70
mptemp,7,80,91,100,110,120,130
mptemp,13,140,150,159,170,179,190
mptemp,19,200,209,220,230,239,250
mptemp,25,260,271,281,291,300
```

Danach erfolgt die Eingabe der spez. Widerstandswerte. Hierbei sind in jeder Anweisung wieder 6 Werte vereinbar. Der zweite Parameter ist die Materialnummer und der dritte die Startnummer.

```
mpdata,rsvx,1,1,0.411,0.262,0.244,0.228,0.216,0.2059
mpdata,rsvx,1,7,0.198,0.192,0.189,0.186,0.185,0.185
mpdata,rsvx,1,13,0.188,0.195,0.210,0.255,1.138,4.690
mpdata,rsvx,1,19,12.877,30.694,71.79,139.52,244.87,405.881
mpdata,rsvx,1,25,628.221,895.32,1178.84,1447.44,1626.511
```

Zur grafische Kontrolle kann eine entsprechende Darstellung organisiert werden. Abb. 1.5-2 zeigt das Ergebnis.

```
/yrange,0.1,1 !Wertebereich der Y-Achse
!/yrange
/axlab,x,Temperatur
mplot,rsvx,1
```




Abb. 1.5-2 spezifischer elektrischer Widerstand als Funktion der Temperatur

Üblich ist auch eine Eingabe über eine Näherungsfunktion.

Die Geometrie des Modells ist einfach ein Rechteck, das im einfachsten Fall mit nur einem Element vernetzt werden kann, wenn die Temperaturunterschiede innerhalb der Pille sehr gering sind. (Wird eine solche Pille aber als Heizer für ein bestimmtes Objekt benutzt, muss auch das thermisch angekoppelte Objekt mit modelliert werden. Da die thermische Kopplung in der Regel dann nicht allseitig erfolgt, kann eine dichte Vernetzung in Wärmeflussrichtung notwendig werden.)

```
!-----Geometrie-----
rectng,0,radpil,0,dipil

!-----Vernetzung-----
esize,,3
amesh,1
```

Die Wärmeabgabe an die Umgebung erfolgt hier durch Konvektion an die Luft. Dazu wird an allen äußeren Randknoten die Konvektionsbedingung vorgegeben.

```
!-----Randbedingungen-thermisch-----
t_umg=20 !Umgebungstemperatur
nselect,s,loc,x,radpil
nselect,a,loc,y,0
nselect,a,loc,y,dipil
sf,all,conv,10,t_umg !Konvektion
allsel
ic,all,temp,t_umg
```

Zur Vorgabe der Spannungsspeisung werden die Knoten der oberen Fläche hinsichtlich des Potentials gekoppelt und ein Knoten für die Einspeisung herausgesucht. Die unteren Knoten werden auf Massepotential (Potential 0 V) gelegt.

```

!-----Randbedingungen-elektrisch-----
nset,s,loc,y,dipil
cp,1,volt,all
*get,n_ein,node,,num,min    !Einspeiseknoten
nset,s,loc,y,0
d,all,volt,0

```

Neben der Vorgabe der Spannung muss für die Auswertung auch eine Möglichkeit geschaffen werden, den Strom zu erfassen. Das geht sehr einfach unter Verwendung eines kleinen Vorwiderstandes. Zusätzlich kann damit auch der Innenwiderstand der Spannungsquelle modelliert werden. Für die Positionierung des Widerstandes wird ein zusätzlicher Knoten benötigt.

```

!-----Speisung über Vorwiderstand-----
u1=8                                !Versorgungsspannung
r1=0.1e-3                            !Vorwiderstand
et,3,circul24,0                       !Widerstand
r,3,r1
abst=2e-3                             !für Knotendarstellung
*get,n_max,node,,num,max              !höchste Knotennummer
n,n_max+1,radpil+abst,abst           !Generierung eines Knoten
type,3 $real,3
e,n_max+1,n_ein                       !Vorwiderstand
*get,e_max,elem,,num,max              !Elementnummer des Vorwiderst.
!-----
d,n_max+1,volt,u1                     !Einspeisung

```




Abb. 1.5-3 Elemente des Modells

Nach Beendigung der Modellerstellung kann die Lösung angefordert werden. Hier soll eine transiente Lösung für den Bereich bis 5 Sekunden ermittelt werden.

```

!-----Lösung-----
allsel
/solu
zeitende=5

```

```

schritte=20

antype,trans
outres,all,all
autots,on          !automatische Schrittweitensteuerung
nsubst,100,200,schritte !Anfangs-, Maximal-, Minimal-
kbc,1
time,zeitende
solve

```

Für eine erste Auswertung wird die Temperatur zum Berechnungsende betrachtet.

```

!-----Auswertung-----
/post1
plnsol,temp

```

Wegen der geringen Wärmeableitung ist sie nahezu homogen und damit wenig aussagefähig. Aussagefähiger sind die Werte für Strom, Widerstand und Temperatur in Abhängigkeit von der Zeit. Dafür wird der Zeitprozessor `/post26` bemüht.

```

/post26
nsol,2,n_ein,temp      !Temperatur
esol,3,e_max,,smisc,2,Strom
!----Ermittlung des Widerstandes----
nsol,4,n_ein,volt     !Spannung
quot,5,4,3,,Widerst  !Spannung/Strom

/grtyp,2              !mehrere Kurven im Diagramm
/yrange,0,200,1      !für Temperatur
/yrange,0,1,2        !für Strom
/yrange,0,200,3      !für Widerstand
/gmarker,2,1         !Stromverlauf mit Dreiecken
/axlab,x,Zeit in s
plvar,2,3,5

```




Abb. 1.5-4 Temperatur, Strom und Widerstand über der Zeit

Um sich mit der Wirkungsweise solcher PTC-Elemente vertraut zu machen, sollten nun verschiedene Durchläufe mit jeweils anderen Speisespannungen gestartet werden. Welche Auswirkungen haben Änderungen der Abmessungen?

(Aus dem Verlauf des Widerstandes ist erkennbar, dass die Vereinbarung des spezifischen Widerstandes als Funktion der Temperatur (vielleicht auch durch Einfügen zusätzlicher Tabellenpunkte zur Homogenisierung) verbesserungswürdig ist.)